/**
 * Copyright 2009 cestum.org
 *
 * Licensed under the Apache License, Version 2.0 (the "License");
 * you may not use this file except in compliance with the License.
 * You may obtain a copy of the License at
 *
 *      http://www.apache.org/licenses/LICENSE-2.0
 *
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 */
package org.cestum.cave.presenter.wave.cursors
{
	import org.waveprotocol.wave.model.document.operation.AnnotationBoundaryMap;
	import org.waveprotocol.wave.model.document.operation.Attributes;
	import org.waveprotocol.wave.model.document.operation.BufferedDocOp;
	import org.waveprotocol.wave.model.document.operation.DocInitializationCursor;
	import org.waveprotocol.wave.model.document.operation.impl.DocOpBuilder;
	
	public class UndoLastLineDocCursor implements DocInitializationCursor
	{
		var lineDeletion:DocOpBuilder = new DocOpBuilder();
		var currLine:int = -1;
		var lastLine:int;
		public function UndoLastLineDocCursor(_line:int)
		{
			lastLine = _line;
		}
		
		public function annotationBoundary(map:AnnotationBoundaryMap):void
		{
		}
		
		public function characters(chars:String):void
		{
			if (currLine == lastLine) {
				lineDeletion.deleteCharacters(chars);
			} else {
				lineDeletion.retain(chars.length);
			}
		}
		
		public function elementStart(type:String, attrs:Attributes):void
		{
			if (type == caveRenderUtils.LINE) {
				currLine++;
			}
			
			if (currLine == lastLine) {
				lineDeletion.deleteElementStart(type, attrs);
			} else {
				lineDeletion.retain(1);
			}
		}
		
		public function elementEnd():void
		{
			if (currLine == lastLine) {
				lineDeletion.deleteElementEnd();
			} else {
				lineDeletion.retain(1);
			}
		}
		
		public function getDocOp():BufferedDocOp {
			return lineDeletion.build();
		}
	}
}